QT += core
TARGET = csv
TEMPLATE = lib

# Uncomment this setting if you want to build static library
#CONFIG += staticlib

!msvc {
    # flags for gcc-like compiler
    CONFIG += warn_on
    QMAKE_CXXFLAGS_WARN_ON += -Werror -Wformat=2 -Wuninitialized -Winit-self \
            -Wswitch-enum -Wundef -Wpointer-arith \
            -Wdisabled-optimization -Wcast-align -Wcast-qual
}

CONFIG(staticlib): DEFINES += QTCSV_STATIC_LIB
DEFINES += QTCSV_LIBRARY

# Uncomment this settings if you want to manually set destination directory for
# compiled library
#CONFIG(release, debug|release): DESTDIR = $$PWD
#CONFIG(debug, debug|release): DESTDIR = $$PWD



HEADERS += \
    abstractdata.h \
    contentiterator.h \
    filechecker.h \
    qtcsv_global.h \
    reader.h \
    stringdata.h \
    symbols.h \
    variantdata.h \
    writer.h

SOURCES += \
    contentiterator.cpp \
    reader.cpp \
    stringdata.cpp \
    variantdata.cpp \
    writer.cpp
